


[bookmark: _GoBack]Fungsi terbilang digunakan untuk membaca suatu bilangan atau nominal uang. Saya  ambil contoh didalam pembuatan kwitansi pastinya ada nominal dan terbilang. Misal di kwitansi tertulis nominal 11.500 dan terbilang akan di tulis “Sebelas ribu lima ratus”. Pada kasus ini bagaimana cara membuat fungsi terbilang ini di dalam suatu program komputer untuk membaca suatu nilai bilangan atau nilai nominal uang. Kasus ini saya ambil berdasarkan pengalaman saya pada saat mengerjakan program atau system administrasi keuangan di suatu daerah, dimana system tersebut harus bisa menampilkan pembacaan jumlah nominal uang.
Langsung saja kita membuatnya. Disini saya mencoba membuat fungsi terbilang tersebut dengan menggunakan Visual Basic 6.0. pertama-tama buka Visual Basic 6.0, sesudah itu langkah berikutnya mendesain terlebih dulu interface nya atau desain programnya. Saya berikan contoh desain programnya sebagai berikut

Komponen dan properti yang digunakan :	
	Komponen
	Properti

	Label1
	· Caption  : Inputkan Nilai

	Label2
	· Caption  : Terbilang

	Text1
	· Name     : Text1
· Text        :

	RichTextBox1
	· Name     :  RichTextBox1
· Text        :

	CommandButton1
	· Name     : Command1
· Caption  : Proses

	CommandButton2
	· Name     : Command2
· Caption  : Clear

	CommandButton3
	· Name     : Command3
· Caption  : Close



Setelah selesai mendesain, selanjutnya kita akan menuliskan coding programnya. disini kita membutuhkan sebuah module untuk di tuliskan sebuah function atau fungsi terbilang yang nantinya akan dipanggil di dalam event yang ada di Command1. Untuk membuat sebuah module bisa klik kanan pada nama projeknya selanjutnya pilih Add => Module. Setelah module dibuat ketikan kode program berikut ini!
	Public Function KeKata(ByVal n As Double)
Dim t As String
Dim Angka() As String

Angka = Split(",satu,dua,tiga,empat,lima,enam,tujuh,delapan,sembilan,sepuluh,sebelas", ",")

If n < 12 Then
    t = " " & Angka(n)

ElseIf n < 20 Then
    t = KeKata(n - 10) & " belas"

ElseIf n < 100 Then
    t = KeKata(n \ 10) & " puluh" & KeKata(n Mod 10)

ElseIf n < 200 Then
    t = " seratus" & KeKata(n - 100)

ElseIf n < 1000 Then
    t = KeKata(n \ 100) & " ratus" & KeKata(n Mod 100)

ElseIf n < 2000 Then
    t = " seribu" & KeKata(n - 1000)

ElseIf n < 1000000 Then
    t = KeKata(n \ 1000) & " ribu" & KeKata(n Mod 1000)

ElseIf n < 1000000000 Then
    t = KeKata(n \ 1000000) & " juta" & KeKata(n Mod 1000000)

ElseIf n < 1000000000000# Then
    t = KeKata(Fix(n / 1000000000)) & " milyar" & KeKata(n - (Fix(n / 1000000000) * 1000000000))

ElseIf n < 1E+15 Then
    t = KeKata(Fix(n / 1000000000000#)) & " trilyun" & KeKata(n - (Fix(n / 1000000000000#) * 1000000000000#))
End If

KeKata = t
End Function

	
Public Function Terbilang(ByVal nilai As Double, Optional ByVal AwalKapital As Boolean = False)
Dim s As String, n As Double, d As String

d = Mid(5 / 2, 2, 1)
n = Abs(nilai)
s = KeKata(n)

If nilai < 0 Then
    s = "minus " & s
End If

s = Trim(Replace(s, "  ", " "))

Terbilang = IIf(AwalKapital = True, StrConv(s, 3), s)
End Function



Setelah kode function ditulis dengan benar dan dipastikan tidak ada yang error, selanjutnya function tersebut kita panggil ke command1. Klik 2x pada bagian Command1 dan isikan kode program berikut ini
	Private Sub Command1_Click()
If Text1.Text = "" Then
    MsgBox "Nilai bilangan belum diinputkan", vbInformation, "Info"
    Text1.SetFocus
    
ElseIf Not IsNumeric(Text1.Text) Then
    MsgBox "Nilai yang diinputkan salah, inputkan dengan type numeric", vbExclamation, "Warning"
    Text1.Text = ""
    Text1.SetFocus
    
Else
    RichTextBox1.Text = Terbilang(Text1.Text)
End If
End Sub



Selanjut pada Command2, caranya masih sama klik 2x pada bagian Command2 dan isikan kode berikut ini
	Private Sub Command2_Click()
Text1.Text = ""
RichTextBox1.Text = ""
Text1.SetFocus
End Sub



Selanjutnya pada Command3 isikan dengan kode program berikut ini
	Private Sub Command3_Click()
Unload Me
End Sub



Setelah semua kode program di ketik semua dan dipastikan tidak ada yang error, jalankan programnya, dan hasilnya seperti pada gambar dibawah ini

Selanjutanya kita test programnya apakah berjalan dengan benar atau tidak. Inputkan bilangan atau nominal uang pada bagian textbox inputkan nilai, setelah diinputkan nilainya klik proses dan hasilnya seperti gambar dibawah ini
[image: ]
Untuk menginputkan bilangan baru klik Clear untuk membersihkan.
Itulah cara membuat fungsi terbilang ini di dalam suatu program komputer untuk membaca suatu nilai bilangan atau nilai nominal uang dengan menggunakan Visual Basic 6.0. semoga tutorial ini bisa bermanfaat bagi mereka yang baru belajar pemrograman khususnya Visual Basic 6.0

